

Standardizing Procedures for Generating Dose Proportionality Figures to Improve Programming Efficiency

Jianli Ping and Krishna Sivakumar, Gilead Sciences Inc.

ABSTRACT

Dose proportionality figures with selected PK parameters provide an intuitive linearity assessment of PK parameters with dose levels for clinical trials in dose escalation. The figure relating the log of the pharmacokinetic parameter linearly to the log of the dose is commonly used. However, the programming procedures for such figures can be time-consuming and tedious. This paper presents generalized procedures to create a natural logarithmic scale of dose proportionality figures that include observed scatter plots, reference line, regression line, and legend annotation through macro calls using SAS. The described procedures can be applied to different studies by adjusting the parameter calls to increase the programming efficiency and accuracy. Some challenges and proposed solutions will be discussed.

INTRODUCTION

Dose proportionality is one of the essential analysis in the early part of the drug development including dose escalation studies, for predicting the effects of dose adjustments and usually includes a graphical representation of exposure analysis of three or more doses. Numerous publications have compared and illustrated how to select proper statistical methods in responding to different study designs and TA areas (Dunnigan 2005; Sheng et al, 2010). Logarithmic scale figures have been adopted by many PK scientists and statisticians because the nature of common PK parameters and dose amounts show linearity with their logarithmic transformed values (Gough K et al., 1995; Shin et al. 2018).

While logarithmic scale figures have been commonly generated, there are some challenges in generating such figures using SAS program due to possible minor differences in logarithmic transformed values between the values selected for display and the actual values that are transformed with SAS function during data preparation procedures. The Proc format with SAS® can provide some flexibility by selecting a range for each selected level; however, it is a manual process to define formatted values and input corresponding intervals with logarithmic transformed values separately, and could result in missing intervals if formatted values do not match exactly with the interval values. This will also increase the programming time and possible discrepancy.

In this paper, we present standardized procedures that allow SAS programmers to generate dose proportionality figures (x,y-axis ranges and key statistics) with macro calls that allow dose amounts and PK parameter intervals in their linear scales. A macro can dynamically create formatted values for each call and assign formatted values in a range for each level. If the intervals are not set properly, they could be adjusted easily with a single step. The selection of statistical methodology can be found from statistical textbook or other publications (Gough K et al., 1995; Dunnigan 2005; Sheng et al, 2010; Shin et al. 2018). This paper focuses on macro calls and a macro to define format values based on users' inputs.

METHOD

WORKFLOW OF DOSE PROPORTIONALITY PROGRAMMING




Figure 1 Workflow for Generating Dose Proportionality Figures

The overall programming procedure for generating the dose proportionality figures can be subdivided into 6 steps.

1. Fetch ADaM datasets ADSL and ADPP, based on selection criteria of analysis flags and specified PK parameters as outlined in the SAP.
2. Data manipulation step that will rename variables to fit variable names reserved for and applied in the standardized macros.

3. Determine parameter values such as dose levels, tick values (intervals) of x-/y-axis in response to dose amounts/corresponding selected PK parameters.
4. Generate the figure by calling the macros.
5. Review the figure outputs to check if the dose amounts, PK parameter intervals, and their linear regression and legend are displayed appropriately. It is important to have the PK parameters to cover the actual ranges and with proper intervals to have the figures displayed properly in terms of their locations and intervals.
6. Complete validation.

MAIN MACRO CALL

Once dataset manipulation is completed, users can filter dataset further and define study specified parameter values. For example, 3 groups of figures could be defined with values as showed in the Figure 2. The first call will select subjects from Cohort 1, 2, 3, 4, and 10, and 3 PK parameters AUCinfo, AUClast and Cmax, with 5 dose levels of 15, 50, 300, and 450 mg for SAD part, and corresponding intervals for the 3 selected PK parameters, separated by '|'. The second and the third call are selecting 3 dose levels of 15, 50 and 150 mg for MAD part. While dose levels depend on study design and all dose levels will be presented with x-axis ticks (intervals), those for each PK parameter used for y-axis ticks could be adjusted according to their ranges and the distance from each other. The selection for y-axis is based on the range from the lowest to the highest of the corresponding PK parameter, adjustments with extra spaces on the top and bottom part of the figure for legend display, and proper tick intervals in y-axis. This selection might be repeated few times before the figures can be finalized (Step 3 through 5 showed in Figure 1).

```
%dpplt(outid=%str(a-sad), condd=%str(cohortn in (1,2,3,4,10) and paramcd in ('AUCIFO','AUCLST','CMAX')),
ordall=%str(15 50 150 300 450|2000 10000 50000 100000 250000|2000 10000 50000 100000 250000|
100 500 2000 5000 10000));

%dpplt(outid=%str(c-mad-d1), condd=%str(cohortn in (6,7,8) and paramcd in ('AUCINT','C24','CMAX') and
avisitn=1), ordall=%str(15 50 150|2000 5000 10000 20000 50000 100000|50 100 500 1000 2000 4000|
100 500 1000 2000 4000 8000));

%dpplt(outid=%str(c-mad-d10), condd=%str(cohortn in (6,7,8) and paramcd in ('AUCTAU','CTAU','CMAX') and
avisitn=10), ordall=%str(15 50 150|2000 5000 10000 20000 50000 100000|100 500 1000 2000 4000 8000|
100 500 1000 2000 4000 8000));
```

Figure 2. Main macro call to specify parameter values for 3 groups of dose proportionality figures

MAIN PROGRAM DEFINITION

The macro 'dpplt' defines the commonly used macro variables, dataset subset, and formats for x-/y-axis, and passed macro values to main macro call as shown in Figure 3. If in need, more macro variables can be defined in the macro header and passed by means of the main macro call. The macro m_pk_dosp within the main macro call will generate the dose proportionality figures.

```

%macro dpplt(outid=, condd=, ordall=);
  data final; set adpp; where &condd; run;

  proc sql; select distinct param into:ylabl separated by '|' from final order by paramcd; quit;
  proc sql; select count(distinct paramcd) into:_xyordn from final; quit;

  %let xyordn=%eval(&_xyordn)+1;
  %put ylabl=&ylabl xyordn=&xyordn;

%dosepfmt;

%m_pk_dosp(outname =%str(g-dosep-&outid),
  titlekey = %str(g-dosep-&outid),
  analfile = final,
  effect = trtan,
  byvar = paramcd,
  xvar = lxvar,
  xlabel = %str(Dose (mg)),
  xfmt = dosffmt.,
  xorder = %str(&xordnew.),
  yorder = %str(&yordnew.),
  ylabel = %str(&ylabl),
  yfmt = dosffmt.,
  spread = 0,
  yvar = lyvar,
  ylvar = refval,
  y2var = regval,
  annot = N,
  ghw = 5.5 9 5 9,
  legendpos = %str(location=inside position=bottomright across=1),
  wantintext = Y,
  vdata = vdata
  );
%mend dpplt;

```

Figure 3 Main program to define and pass macro values to generate the dose proportionality figure

MACRO TO GENERATE STUDY SPECIFIED FORMAT

To generate format and intervals for x-/y-axis more effectively, a macro named 'dosepfmt' is defined and called, which will generate a specified format that consolidate all x-/y-axis intervals with selected values in the main macro call. To make sure each selected interval is displayed properly, a range will be defined

	LABEL	START	END	FMTNAME
1	15	2.7079	2.7083	dosffmt
2	50	3.9118	3.9122	dosffmt
3	100	4.605	4.6054	dosffmt
4	150	5.0104	5.0108	dosffmt
5	300	5.7036	5.704	dosffmt
6	450	6.109	6.1094	dosffmt
7	500	6.2144	6.2148	dosffmt
8	2000	7.6007	7.6011	dosffmt
9	5000	8.517	8.5174	dosffmt
10	10000	9.2101	9.2105	dosffmt
11	50000	10.8196	10.82	dosffmt
12	100000	11.5127	11.5131	dosffmt
13	250000	12.429	12.4294	dosffmt

	LABEL	START	END	FMTNAME
1	15	2.7079	2.7083	dosffmt
2	50	3.9118	3.9122	dosffmt
3	100	4.605	4.6054	dosffmt
4	150	5.0104	5.0108	dosffmt
5	500	6.2144	6.2148	dosffmt
6	1000	6.9076	6.908	dosffmt
7	2000	7.6007	7.6011	dosffmt
8	4000	8.2938	8.2942	dosffmt
9	5000	8.517	8.5174	dosffmt
10	8000	8.987	8.9874	dosffmt
11	10000	9.2101	9.2105	dosffmt
12	20000	9.9033	9.9037	dosffmt
13	50000	10.8196	10.82	dosffmt
14	100000	11.5127	11.5131	dosffmt

Figure 4 Formats generated with macro 'dosepfmt' for SAD (left) and MAD, Day 1 group (right)

in START/END with logarithmic transformed values with the label in their linear scale (Figure 4). Since SAD and MAD parts have different dose amounts and corresponding PK parameter ranges, the format values are different per each call and each PK parameter, with 13 and 14 levels for AUCinf in SAD and MAD respectively. The START/END range can ensure each expected interval/tick to be displayed in the figures. Without this macro, the format will need to be defined separately and assigned manually into main macro call with logarithmic transformed values.

SAS CODE IN THE MACRO TO GENERATE STUDY SPECIFIED FORMAT

The detailed SAS code to automatically generate the formats for x-/y-axis defined as macro 'dosepfmt' is provided below. This macro is mainly composed of three sections.

- In the first section, it collects all x-/y-axis intervals as defined in the main macro call based on the total PARAMCD (PK parameter) selected, takes logarithmic values rounding to 0.0001, assigns END/START by +/- 0.0002, and keeps only unique intervals.
- In the second section, it gets the linear values for each x-/y-axis as tick label and actual logarithmic START/END values to be pulled in the main figure macro.
- In the third section, it dynamically generates format, and remove temporary format values to allow next call to be refreshed.

```
%macro dosepfmt;
  %global xordnew yordnew;

  /*SECTION 1*/
  %do i=1 %to &xyordn;
    %let ord&i=%scan(&ordall, %eval(&i.), '|');

    data fmt_&i.;
      length tempstr $500 lnvalc $20;
      tempstr="&&ord&i.";
      do j=1 to countw(tempstr);
        origval=input(scan(tempstr,j," "),best.);
        groupord=&&i.;
        if origval>0 then lnval=round(log(origval), 0.0001);
        s_lnval=lnval-0.0002;
        u_lnval=lnval+0.0002;
        lnvalc=strip(put(lnval, best.));
        output;
      end;
    run;

    proc sql; select distinct lnval into :ordlabl_&i. separated by ' ' from
      fmt_&i. order by lnval; quit;
    %put _user_;

    proc append base=fmtall data=fmt_&i. force;
    run;
  %end;

  proc sort data=FMTALL; by groupord j; run;
```

```

proc transpose data=fmtall out=ordall_tr prefix=n_;
  by groupord;
  id j;
  var lnvalc;
run;

proc sql; select max(j) into:maxcoln from fmtall; quit;

data ordall_tr_n;
  set ordall_tr;
  length ordlabl $500;
  ordlabl=catx(" ", of n_1-n_%eval(&maxcoln.));
run;

/*SECTION 2*/

proc sql; select distinct ordlabl into:xordnew
  from ordall_tr_n
  where groupord=1;
quit;
proc sql; select distinct ordlabl into:yordnew separated by '|'
  from ordall_tr_n
  where groupord^=1
  order by groupord;
quit;

proc sort data=fmtall out=fmtall_final_0 nodupkey;
  by origval;
run;

data fmtall_final;
  set fmtall_final_0;
  by origval;
  retain fmtname "dosffmt";
  rename s_lnval=start u_lnval=end origval=label;
run;

/*SECTION 3*/

proc format cntlin=fmtall_final fmtlib;
  select dosffmt;
  title "Dose proportionality format";
run;

proc datasets library=work;
  delete fmtall;
quit;
%mend dosepfmt;

```

SAS CODE IN THE MACRO TO GENERATE THE PLOT

The macro %m_pk_dosp is called to generate the figure, including scatter plots of dose amounts and selected PK parameter values, their regression line, key statistics for the regression model, and legend.

The 2 main sections of the macro are explained below:

- In the 1st section, mean dose amount (mlogx) and mean PK parameter (mlogy) with Proc means are generated to be used for calculating the reference line. The regression modeling is determined by using Proc mixed. The selected regression parameters are displayed in the plot legend dynamically.
- In the 2nd section, Graph Template Language (GTL) is used to create and overlap the scatter plot, regression line, reference line and legend together. %do loop allows all PK parameters in each group such as same part and same visit to be displayed as one page per PK parameter and an output file for all selected PK parameters.

```
/*SECTION 1*/
proc means data=&analfile n mean nway;
  class grpn avisitn paramcd;
  var lyvar lxvar;
  output out=gdata_sum(drop=_) n=nlogy nlogx mean=mlogy mlogx;
run;

proc mixed data=&analfile;
  class usubjid;
  by grpn avisitn paramcd;
  model lyvar=lxvar / ddfm=kr;
  estimate "Slope of log dose" lxvar 1 / cl alpha=0.10;
  make 'estimates' out=slope1;
run;

data gdata;
  merge &analfile slope1 gdata_sum;
  by grpn avisitn paramcd;
  reflny=1*lxvar+(mlogy-mlogx);
run;

data legendval;
  set slope1;
  by grpn avisitn paramcd;
  retain cnt;
  length legendval $150;
  if first.avisitn then cnt=1;
  else cnt+1;
  legendn=grpn||compress(avisitn)||compress(cnt);
  legendval="Regression line based on the power model%str(;) Estimated slope
and 90% CI = "||put(estimate, 4.2)||' ('||put(lower, 4.2)||', '||put(upper,
4.2)||')';
run;

/*SECTION 2*/
proc sort data=&analfile out=figcd nodupkey;
  by paramcd;
run;
```

```

data _null_;
  set figcd;
  by paramcd;
  call symput("max",compress(put(_n_,8.)));
  call symput("paramcd_"||compress(put(_n_,8.)),compress(paramcd));
run;

data _null_;
  set legendval;
  call symput(legendn,legendval);
run;

%do i=1 %to &max;
  %if &i=1 %then %do; %let gsfmode=replace; %end;
  %else %do; %let gsfmode=append; %end;

proc template;
  define statgraph myplot;
  begingraph;
  entrytitle "";
  layout overlay /
    xaxisopts=(&xordnew) label="&xlabel" inearopts=(tickvaluelist=(&xordnew)
      tickvalueformat=&xfmt)
    yaxisopts=(&yordnew) label="&ylabel" inearopts=(tickvaluelist=(&yordnew)
      tickvalueformat=&yfmt)
    y2axisopts=(&yordnew) label="" linearopts=(tickvaluelist=(&yordnew)
      tickvalueformat=&yfmt);
  scatterplot y=lyvar x=lxvar / markerattrs=(symbol=square size=10px);
  seriesplot y=reflny x=lxvar / name='ref'
    lineattrs=GraphFit(pattern=shortdash color=black)
  legendlabel="Reference line (45 degree line at mean)";
  regressionplot y=lyvar x=lxvar /name='reg' lineattrs = (pattern=solid
    color=black)
  legendlabel="%sysfunc(cats(%nrstr(&),&grpnr, &visn, &i))";
  discretelegend 'reg' 'ref'/location=inside halign=right valign=bottom;
  endlayout;
  endgraph;
end;
run;

proc sgrender data=&analfile(where=(paramcd="&&paramcd_&i"))
  template='myplot';
run;


%end;

```

OUTPUT


Running the macro program will generate 3 groups of dose proportionality figures (total 9), with one figure for each pair of dose amounts and PK parameter, since 3 PK parameters have been defined for each main macro call. Figure 5 is the first figure output named 'g-dosep-a-sad' reflecting relationship between 5 dose amounts and AUCinf with ranges from 2000 through 250000 ng/mL. The legend is generated

dynamically with the macro call and displays the estimated slope, slope confident intervals, and 45° reference line. Figure 6 is the first figure named 'g-dosep-a-mad' reflecting relationship between dose amounts and AUCinf with ranges from 2000 through 100000 ng/mL.



SAD = single ascending doses.
Source: Table 15.10.1.1.2.1.

Figure 5. First figure in 'g-dosep-a-sad' shows relationship between dose and AUCinf



MAD = multiple ascending doses.
Source: Table 15.10.1.1.2.3.

Figure 6. First figure in 'g-dosep-a-mad' shows relationship between dose and AUCinf

The tick numbers and values in the y-axis can be easily adjusted by either adding/removing or setting different values in the main macro call based the initial outputs since initial selection might be too far/close or not marked evenly.

DISCUSSION

As shown above, the macros 'dosefmt' and m_pk_dosp, will help to improve programming efficiency, flexibility, and accuracy in the generation of the dose proportionality figures. The format updates are performed programmatically by adjusting the selected dose amounts and PK parameter tick values in the main macro calls. For manual format selection, users must find each unique value to be included in the figures, determine its logarithmic value, and add new value into the proc format list and main macro call separately. There is a possibility of missing ticks if selected logarithmic values are not close enough to the value calculated from the SAS function. Although users could add a range such as 1000, shown in Figure 7, it would require updates in multiple locations when ranges are changed. The proposed macro can help select the linear values, and a simple rerun can help choose the START/END range in the log-transformed values after the main macro call is updated. The macro m_pk_dosp can have major components of dose-proportionality figures created dynamically with selected statistics included in the legend.

```
proc format;
  value figfmt
    2.7080502='15'
    3.912023='50'
    4.60517='100'
    5.010635294='150'
    5.70378247='300'
    6.10924758='450'
    6.214608='500'
    6.907754-6.907756='1000'
    7.600902='2000'
    8.0063='3000'
    8.2940496='4000'
    8.51719319='5000'
    8.5171-8.5173='5000'
    8.699514748='6000'
    8.853665428='7000'
    8.98719682='8000'
    9.21034='10000'
    9.903488='20000'
    10.819778='50000'
    11.512925='100000'
    11.918391='150000'
    12.20607='200000'
    12.429216='250000'
;
run;
```

```
%doit(outid=%str(a-sad), condd=%str(cohortn in (1,2,3,4,10) and paramcd in ('AUCIFO','AUCLST','CMAX')),
  xord=%str(2.7080502 3.912023 5.010635294 5.70378247 6.10924758), ylabl=%str(AUCinf (h*ng/mL)|AUClast (h*ng/mL)|Cmax (ng/mL)),
  yord=%str(7.600902 8.517193 9.21034 9.903488 10.819778 11.512925 12.429216|7.600902 8.517193 9.21034 9.903488
    10.819778 11.512925 12.429216 |4.60517 6.214608 7.600902 8.51719319 9.21034));
```

Figure 7 Initial format values and selected main macro call

CONCLUSION

This paper presents standardized SAS codes through 2 macros to generate dose proportionality figures. Compared to initial procedures, the described methods can increase programming efficiency by directly adjusting the parameters to the macro calls. The START/END range dynamically generated with the macro for each selected PK parameter can ensure each selected point can be displayed as expected.

REFERENCES

Dunnigan, K 2005. Using SAS to Assess Dose Proportionality in Dose Escalation Studies. MWSUG 2005 Conference Proceedings. Available at <https://www.semanticscholar.org/paper/PH-300-Using-SAS-to-Assess-Dose-Proportionality-in-Dunnigan-Louis/c6c9d1e26f14c098d883e4eca40d607d4713239e>

Sheng Y, He Y, Huang X, Yang, J, Wang K, and Zheng Q. 2010. Systematic Evaluation of Dose Proportionality Studies in Clinical Pharmacokinetics. *Current Drug Metabolism*, 2010, Vol. 11, No. 6, 526-537.

Shin KH, Jang JK, Kim K, and Kim MG. 2018. Dose-proportional pharmacokinetic properties of GLA 5PR GLARS -NF1 controlled-release pregabalin in healthy Korean volunteers: a randomized, open, single-dose, parallel study. *Drug Design, Development and Therapy* 2018:12 3449–3457.

Gough K, Hutchison M, Keene O, Byrom B, Ellis S, Lacey, L, and McKellar J. 1995. Assessment of Dose Proportionality: Report from the Statisticians in the Pharmaceutical Industry/Pharmacokinetics UK Joint Working Party. 29(3): 1039-1048. Available at <https://doi.org/10.1177/009286159502900324>

ACKNOWLEDGEMENTS

The authors would like to thank the encouragement and support from Phase 1 Programming Team and Programming Department.

CONTACT INFORMATION

Jianli Ping
Gilead Sciences, Inc.
333 Lakeside Drive Foster City, CA 94404 USA
Email: Jianli.Ping@gilead.com

Krishna Sivakumar
Gilead Sciences, Inc.
333 Lakeside Drive Foster City, CA 94404 USA
Email: Krishna.Sivakumar@gilead.com